


OCaPIC tutorial  
Version 1.2

# Programming PIC microcontrollers with OCaml

---

Benoît Vaugon



April 19, 2012

*English translation: Zoé Faget*

# 1 Introduction

OCaPIC is a set of programs and libraries allowing programming PIC microcontrollers in OCaml. All OCaml features are available on PIC. OCaPIC provides the major parts of the standard OCaml library as well as specific PIC libraries allowing direct interaction with hardware interruption, LCD display, serial connection. . .

Furthermore, OCaPIC holds a simulation mechanism in which one can improve an OCaml program before transferring it on the PIC.

For further information, please refer to OCaPIC's website

<http://www.algo-prog.info/ocapic/>

## 2 Installation

### 2.1 Prerequisites

- Any C-compiler (for example `gcc`).
- The standard OCaml compiler provided by INRIA (<http://ocaml.org/>). On most Linux distribution it is sufficient to install the `ocaml` package.
- A PIC assembler in order to convert OCaPIC generated assemblers files (`.asm`) into hexadecimal code (`.hex`), to be transferred on the PIC. The `gpasm` program which comes with the `gputils` Linux package is a possible option.

### 2.2 Downloading

From the OCaPIC website: <http://www.algo-prog.info/ocapic/>, *Download* section, download the latest available version. You should get a file named `ocapic-X.Y.tar.bz2` (replace `X.Y` by the version number).

### 2.3 Installation

Here is a guideline to install OCaPIC on Linux. The Windows installation is also possible thanks to Cygwin.

In a terminal window (each time replacing `X.Y` by the version number):

- Extract the downloaded file `ocapic-X.Y.tar.bz2`:

```
tar jxf ocapic-X.Y.tar.bz2
```

- Go to the `ocapic-X.Y` directory:

```
cd ocapic-X.Y/
```

- Configure OCaPIC. This operation checks dependencies and allows to configure installation directories and managed PIC. Only `PIC18F4620` is managed by default, because the compilation of the interface with each PIC is very long. For more details please refer to the `INSTALL` file:

```
./configure
```

- Compile OCaPIC :

```
make
```

- Install OCaPIC (as super-user) :

```
sudo make install
```

Or:

```
su
```

```
make install
```

## 3 Use examples

All examples featured in this tutorial are adapted to the PIC18F4620. Although all PIC of the 18F serie are supported, we recommend that you start with this one.

### 3.1 Blink a LED

We begin with a very simple example: how to blink a LED with a PIC. There is of course very little interest in using a high-level language such as OCaml to do so, nor a microcontroller. The purpose of this section is to highlight the different steps of PIC programming in OCaml.

#### OCaml source code:

Place the following OCaml code in a `led.ml` file:

```
1 open Pic;; (* Module containing write_reg, set_bit, RB0, ... *)
2
3 write_reg TRISB 0x00; (* Configure the B port as output *)
4
5 while true do
6   set_bit RB0; (* Turn on the LED *)
7   Sys.sleep 500; (* Wait 0,5s *)
8   clear_bit RB0; (* Turn off the LED *)
9   Sys.sleep 500; (* Wait 0,5s *)
10 done
```

#### PIC configuration:

It is necessary to specify several of PIC's configuration options (programming voltage, clock selection, external reset configuration, protected code segments selection, ...). To do so, place the following code into a `config.asm` file:

```
1 config OSC = INTIO7
2 config PWRT = ON
3 config BOREN = OFF
4 config WDT = OFF
5 config MCLRE = OFF
6 config PBADEN = OFF
7 config STVREN = OFF
8 config LVP = OFF
```

#### Compilation:

Compile the `led.ml` file for PIC18F4620 with the following command:

```
ocapic 18f4620 config.asm led.ml
```


You should get, among others, the `led.asm` files (containing PIC assembler code) and `led` (executable file used for simulation).

#### Simulation:

Simulate the execution of your program on PIC with the following command:

```
./led ocapic_dip40_simulator
```

You should get a window representing a PIC with a blinking B0 pin:



Since the program does not end (`while true`), kill it in the terminal window with `Ctrl-C`.

**Assembling:**

If you're using `gpasm`, assemble the `led.asm` file with:

```
gpasm -y led.asm
```

This command generates, among others, the hexadecimal file `led.hex` to be transferred to the microcontroller.

Remark : the `-y` option enables PIC's "extended instructions set". Make sure to enable this option if you use another assembler than `gpasm`.

**PIC programming:**

You now have a `led.hex` file which can be directly transferred to the PIC through a "PIC programmer". If you are lucky enough to have a computer with serial port, you may use a low cost "serial programmer" (or build one yourself with a few components). Otherwise, use a USB programmer. The "USBPICPROG" project (<http://usbpicprog.org/>) helps getting and making one at a reasonable price.

**Circuit:**

Connect a LED between the PIC's B0 pin and the - of the power supply (through a resistance, of course !), supply the PIC with 5V and you're done !

## 3.2 Hello world on an LCD display

OCaPIC provides a library named `Lcd` which allows to operate on small LCD displays with a PIC. For complete documentation, please refer to the corresponding manpage. Here is a short example:

### OCaml source code :

Place the following OCaml code in a `hw.ml` file:

```
1 open Pic;;
2
3 module Disp = Lcd.Connect (
4   struct
5     let bus_size = Lcd.Four          (* LCD/PIC connection configuration: *)
6     let e = LATD0                   (* Bus size *)
7     let rs = LATD2                  (* Validation pin: D0 *)
8     let rw = LATD1                  (* Instruction/data pin: D2 *)
9     let bus = PORTB                 (* Read/write pin: D1 *)
10  end
11 );;
12
13 Disp.init ();;                      (* Display initialization *)
14 Disp.config ();;                    (* Display configuration *)
15 Disp.print_string "Hello_world";;  (* Display a string *)
```

### PIC configuration:

As in section 3.1, place the following code of configuration in a `config.asm` file:

```
1   config OSC      = INTIO7
2   config PWRT     = ON
3   config BOREN    = OFF
4   config WDT      = OFF
5   config MCLRE    = OFF
6   config PBADEN   = OFF
7   config STVREN   = OFF
8   config LVP      = OFF
```

### Compilation:

Compile `hw.ml` → `hw, hw.asm` :

```
ocapic 18f4620 config.asm hw.ml
```

Assemble `hw.asm` → `hw.hex` :

```
gpasm -y hw.asm
```

### Simulation:

Execute the following command:

```
./hw 'ocapic_lcd_simulator 16x2 e=RD0 rs=RD2 rw=RD1 bus=PORTB'
```


You should get a window similar to this one:



Since this program does end, to stop the simulation, simply hit `Enter` in the terminal window.

**Circuit:**

Connect an LCD display to the PIC in the following fashion:



Supply the PIC and the LCD display with 5V.

### 3.3 A small timer

This short example aims at showing how to use an interesting plug-in for simulation: the “circuit” simulator. By “circuit” we mean the connection of a set of standard components to the PIC (such as switches, push buttons, LED, ...).

We intend to create here a small *timer* with an LCD display on which runs a time counter, and several push buttons for adjusting time delay and start/stop time-out.

#### OCaml source code:

Place the following code in a `timer.ml` file:

```
1 open Pic;;
2
3 (* PIC clock speed-up *)
4 set_bit IRCF1;; set_bit IRCF0;; set_bit PLEN;;
5
6 (* PIC/LCD display connection configuration *)
7 module Disp = Lcd.Connect (
8   struct
9     let bus_size = Lcd.Four
10    let e = LATD0
11    let rs = LATD2
12    let rw = LATD1
13    let bus = PORTC
14  end
15 );;
16
17 (* Display initialization and configuration *)
18 Disp.init ();; Disp.config ();;
19
20 (* Counter (reference to an Int32) *)
21 let counter = ref 0;;
22
23 (* Display utility *)
24 let print_02d n =
25   if n < 10 then Disp.print_char '0';
26   Disp.print_int n;
27 ;;
28
29 (* Displays counter's value *)
30 let print () =
31   let t = !counter in
32   let t60 = Int32.div t 60 in
33   let s = Int32.to_int (Int32.rem t 60) in
34   let m = Int32.to_int (Int32.rem t60 60) in
35   let h = Int32.to_int (Int32.div t60 60) in
36   Disp.moveto 0 0;
37   print_02d h;
38   Disp.print_char ':';
39   print_02d m;
40   Disp.print_char ':';
41   print_02d s;
42 ;;
43
44 (* Overflow utilities *)
45 let is_null () = !counter <= 01 and is_full () = !counter >= 3599991;;
46
47 (* Increments the counter by n *)
48 let incr n =
49   counter := Int32.add !counter (Int32.of_int n);
50   if is_full () then counter := 3599991;
51 ;;
52
53 (* Decrements the counter by n *)
54 let decr n =
55   counter := Int32.sub !counter (Int32.of_int n);
56   if is_null () then counter := 01;
57 ;;
58
59 (* Increments/Decrements the counter by 1 *)
60 let decr1 () = if not (is_null ()) then counter := Int32.pred !counter;;
61 let incr1 () = if not (is_full ()) then counter := Int32.succ !counter;;
62
```

```

63 (* Speed-up functions for time setting *)
64 let step n = n / 32 * n / 16 * n + n / 16 * n / 16 + n / 8 + 1;;
65
66 (* Small states machine *)
67 let rec run n =
68   if is_null () || test_bit RB0 then stop ()
69   else if test_bit RB1 then ( incr_start () ; run 0 )
70   else if test_bit RB2 then ( decr_start () ; run 0 )
71   else if n = 79 then ( Sys.sleep 7 ; decr1 () ; print () ; run 0 )
72   else ( Sys.sleep 10 ; run (succ n) )
73 and pause () =
74   Sys.sleep 10;
75   if test_bit RB0 && not (is_null ()) then start ()
76   else if test_bit RB1 then ( incr_start () ; pause () )
77   else if test_bit RB2 then ( decr_start () ; pause () )
78   else pause ();
79 and start () =
80   set_bit RB3;
81   Sys.sleep 10;
82   if test_bit RB0 then start () else run 0;
83 and stop () =
84   clear_bit RB3;
85   Sys.sleep 10;
86   if test_bit RB0 then stop () else pause ();
87 and incr_start () =
88   incl ();
89   print ();
90   Sys.sleep 300;
91   incr_loop 1;
92 and decr_start () =
93   decl ();
94   print ();
95   Sys.sleep 300;
96   decr_loop 1;
97 and incr_loop n =
98   if test_bit RB1 then (
99     incr (step n);
100    print ();
101    Sys.sleep 100;
102    if not (is_full ()) then incr_loop (succ n);
103   )
104 and decr_loop n =
105   if test_bit RB2 then (
106     decl (step n);
107     print ();
108     Sys.sleep 100;
109     if not (is_null ()) then decr_loop (succ n);
110   )
111 in
112 write_reg TRISB 0b00000111; (* Configure button pins as input *)
113 print (); (* Display 00:00:00 *)
114 stop (); (* Entry point of the state machine *)

```

### PIC configuration:

Create the same configuration file config.asm as in section 3.1.

### Virtual circuit:

The plug-in which enables circuit simulation takes as parameter a file name containing the circuit description in a rather intuitive format. Place the following definitions in a circuit.txt file:

```

1 window width=270 height=220 title="Timer"
2
3 button x=85 y=40 width=40 color=black label="START" pin=RB0
4 button x=135 y=55 width=20 height=20 label="+" pin=RB1
5 button x=135 y=25 width=20 height=20 label="-" pin=RB2
6
7 led x=175 y=55 pin=RB3 color=green
8 led x=175 y=25 pin=RB3 inverse=true
9
10 lcd x=10 y=100 column_nb=8 line_nb=1 e=RD0 rs=RD2 rw=RD1 bus=PORTC

```

**Compilation:**

Compile `timer.ml` → `timer, timer.asm` :

```
ocapic 18f4620 config.asm timer.ml
```

Assemble `timer.asm` → `timer.hex` :


```
gpasm -y timer.asm
```

**Simulation:**

Execute:

```
./timer 'ocapic_circuit_simulator circuit.txt'
```

You should get a window similar to this one:



You can now simulate a user's interaction with the timer by clicking +, - and START. To stop simulation, hit `Ctrl-C` in the terminal window.

### 3.4 Communication with a computer

Communication between a computer and a PIC can be established thanks to the two modules `Serial` of the library (one for the PIC and one for the computer). They allow the direct exchange of OCaml values between the program running on the computer and the program running on the PIC.

The exchanged OCaml values between the two programs can be of any kind as long as it makes sense. For example those values can be integers, strings or float numbers but also more complex data structures such as lists, arrays, etc. The only forbidden values are those containing code pointers such as closures. It is a true exchange of *memory graph*, values may contain cycles.

Communication is protected from transmission errors, accidental unplugging, etc. Furthermore, memory overflow and collisions in the protocol are correctly managed and translated into exceptions.

If you don't have a serial port on your computer, you may use a USB/Serial converter.

The use of this library is entirely managed by the simulator.

The precise documentation of each function is in manpages.

Here is an example of how to drive an LCD display connected to the PIC from an OCaml top-level running on a computer:

#### OCaml source code for the Computer:

Place the following source code in a `sender.ml` file, adapting if needed the file path corresponding to the serial port (here `/dev/ttyUSB0`):

```
1 open Serial;;
2
3 (* Type of values transfered from the computer towards the PIC *)
4 type t = Clear | Moveto of int * int | Int of int | Text of string
5
6 (* Communication channel with the PIC via the serial port *)
7 let (chan : (t, unit) channel) = open_tty "/dev/ttyUSB0";;
8
9 (* Utilities *)
10 let clear () = send chan Clear;;
11 let moveto l c = send chan (Moveto (l, c));;
12 let print_int i = send chan (Int i);;
13 let print_string s = send chan (Text s);;
```

#### OCaml source code for the Computer (simulation mode):

Place the following code in a `simul.ml` file:

```
1 open Serial;;
2
3 (* Type of values transfered from the computer towards the simulator *)
4 type t = Clear | Moveto of int * int | Int of int | Text of string
5
6 (* Communication channel with the simulator *)
7 let (chan : (t, unit) channel) =
8   open_prog "./usart_/ocapic_lcd_simulator_16x2_e=RD0_rs=RD2_rw=RD1_bus=PORTB"
9 ;;
10
11 (* Utilities *)
12 let clear () = send chan Clear;;
13 let moveto l c = send chan (Moveto (l, c));;
14 let print_int i = send chan (Int i);;
15 let print_string s = send chan (Text s);;
```

### OCaml source code for PIC :

Place the following code in a `usart.ml` file, adapting if needed the given argument to the `open_channel` function (here 34 for 57600 baud) according to the rate of your serial port (please refer to the pic documentation for the full conversion list).

```
1 open Pic;;
2 open Lcd;;
3 open Serial;;
4
5 (* PIC clock speed-up *)
6 set_bit IRCF1;; set_bit IRCF0;; set_bit PLLLEN;;
7
8 (* PIC/LCD display connection configuration *)
9 module Disp = Connect (
10 struct
11   let bus_size = Four
12   let e = LATD0
13   let rs = LATD2
14   let rw = LATD1
15   let bus = PORTB
16 end
17 );;
18
19 (* Initialization and configuration of the display *)
20 Disp.init ();; Disp.config ~cursor:Underscore ();;
21
22 (* Type of values transfered from the computer towards the PIC *)
23 type t = Clear | Moveto of int * int | Int of int | Text of string
24
25 (* Communication channel with the computer *)
26 let (chan : (unit, t) channel) = open_channel 34;; (* 57600 bauds *)
27
28 (* Loop : command reception / transmission to the display *)
29 while true do
30   match receive chan with
31   | Clear -> Disp.clear ()
32   | Moveto (l, c) -> Disp.moveto l c
33   | Int i -> Disp.print_int i
34   | Text t -> Disp.print_string t
35 done
```

### PIC configuration :

Create the same configuration file `config.asm` as in section 3.1.

### Compilation:

Remark : if you have installed the OCaPIC library somewhere else than in `/usr/lib`, modify the access path accordingly.

Compile `sender.ml` → `sender.cmo` :

```
ocamlc -I /usr/lib/ocapic/extra -c sender.ml
```

Compile `simul.ml` → `simul.cmo` :

```
ocamlc -I /usr/lib/ocapic/extra -c simul.ml
```

Compile `usart.ml` → `usart,usart.asm` :

```
ocapic 18f4620 config.asm usart.ml
```

Assemble `usart.asm` → `usart.hex` :

```
gpasm -y usart.asm
```

### Simulation :

Launch the command:

```
ocaml -I +threads -I /usr/lib/ocapic/extra unix.cma threads.cma
serial.cmo simul.cmo
```

This opens an OCaml top-level allowing you to call functions from the `Simul` module which you've just created, as well as a window representing the LCD display connected to the PIC. You then have control over the LCD display by writing commands in the top-level such as:


```
# open Simul;;
# print_int 42;;
# moveto 2 6;;
# print_string "Hello";;
```



To end the simulation, kill the OCaml top-level by hitting `Ctrl-D`.

### Circuit:

Since the logical levels of the PIC (0V/5V) are not the same as those of the serial port (10V/-10V), use the integrated circuit MAX232 to make conversions. The connection to the LCD display is done as in section 3.2 :



### Run :

After transferring the `usart.hex` file on the PIC, connecting the serial port to the circuit and supplying the circuit with 5V, run in the terminal window:

```
ocaml -I +threads -I /usr/lib/ocapic/extra unix.cma threads.cma
serial.cmo sender.cmo
```

You then get a new OCaml top-level with which you can control the LCD display via the `Sender` module. You should get the same behavior as in the simulator, with the difference that now you are operating the real LCD display connected to the PIC !

## 4 Advance use

### 4.1 Virtual machine settings

Several programs may need a setting for the OCaml heap's or stack's size. On the PIC18F4620, default settings are 174 levels in the stack and 1792 bytes for the heap.

Adjustments of the size of the heap and stack takes place during the *link* step when calling `ocapic`. The `ocapic` command has two options `-stack-size` and `-heap-size`. Each one takes as argument the number of 256 bytes segments allocated either to the heap or the stack.

For example, running:

```
ocapic 18f4620 -stack-size 3 -heap-size 6 config.asm hw.ml
```

defines 430 levels of stack and 1536 bytes of heap.

In the following table, we describe all possible configurations allowing the use all the PIC registers for the OCaml's stack and heap. Those values may reduce in case of interfacing with assembler code needing more than 15 bytes of memory. For further details, please refer to section 4.2.

-stack-size	-heap-size	number of stack levels	size of heap (in bytes)
1	7	174	1792
3	6	430	1536
5	5	686	1280
7	4	942	1024
9	3	1198	768
11	2	1454	512
13	1	1710	256

One can note that freeing a heap segment allows to use two for the stack. This is due to the virtual memory manager based on the *Stop & Copy* algorithm, which divides by 2 the heap's usable size.

### 4.2 Binding with assembler code

It is possible to interface OCaml code with PIC assembler code. This assembler code may even have been generated by another compiler, provided we have some control over the labels generated by the compiler.

On the OCaml size, interfacing is done with the keyword `external`. The `external` function's name is then the assembler code's label. The assembler code should be written in one (or several) `.asm` files and given to the `ocapic` command at *link* step.

A quick example:

#### OCaml source code:

Place the following code in a `mask.ml` file:

```
1 external mask_portb : int -> int -> unit = "mask_portb";;
2
3 Pic.write_reg Pic.TRISB 0x00;; (* PORTB as output*)
4 Pic.write_reg Pic.PORTB 0x00;; (* PORTB <- 0 *)
5
6 for i = 1 to 100 do
7   mask_portb i 0xF0;      (* Execute the external code *)
8   Sys.sleep 1000;        (* Waits 1s *)
9 done
```

### Assembler source code:

Place the following external function code `mask_portb` in a `ext.asm` file :

```
1 mask_portb:
2     rrcf    ACCUH, W           ; PORTB <- Long_val(ACCU)
3     rrcf    ACCUL, W
4     movwf   PORTB
5     rrcf    [0x2], W          ; PORTB <- PORTB ^ STACK[0]
6     rrcf    [0x1], W
7     xorwf   PORTB, F
8     clrf   ACCUH              ; return ()
9     movlw   0x1
10    movwf   ACCUL
11    return
```

### C source code:

In order for the simulator to work, the `mask_portb` function's assembler code must be recoded in C. Place the following code in a `ext.c` file:

```
1 #include <caml/mlvalues.h>
2
3 /* Simulator's internal function */
4 /* Writing in a special register of the PIC */
5 value caml_pic_write_reg(value vreg, value vval);
6
7 /* C Version of the mask_portb function */
8 value mask_portb(value v1, value v2){
9     caml_pic_write_reg(Val_long(0x01), Val_long(Long_val(v1) ^ Long_val(v2)));
10    return Val_unit;
11 }
```

### PIC configuration:

Create the same configuration file `config.asm` as in section 3.1.

### Compilation:

Compile `mask.ml`, `ext.c`, `ext.asm` → `mask`, `mask.asm` :

```
ocapic 18f4620 config.asm mask.ml ext.c ext.asm
```

Assemble `mask.asm` → `mask.hex` :


```
gpasm -y mask.asm
```

### Simulation:

Run the following command:

```
./mask 'ocapic_dip40_simulator'
```

You should get a window looking like this:



### 4.3 Interruption catching

As we can see in the Pic module's manpage of the standard library, two functions allow to operate on the PIC's hardware interruptions from OCaml :

```
1 (* Registers the interruption handler *)
2 val set_interruption_handler : (bit -> unit) -> unit
3
4 (* Removes the interruption handler *)
5 val clear_interruption_handler : unit -> unit
```

Catching up those interruptions is similar to catching up Unix signals. The function `set_interruption_handler` allows to register an OCaml function of type `bit -> unit`. This *handler* is called each time an interruption occurred with as argument the special register's bit corresponding to the interruption. You can refer to the PIC's documentation to know the correspondance between interruptions and bits. For example, the bit matching the interruption generated by a state change for the B0 pin is `INTOF`.

As for the standard management of Unix signals in OCaml, an interruption can not be managed while an external function is running. Therefore, the wait before an interruption is managed is not equal to a machine cycle (as it is in most cases when coding in assembler for PIC), but is equal to an OCaml virtual machine cycle. This may cause reactivity problems in case of long external calls, such as a call to the `Sys.sleep` function.

Moreover, an interruption management can be interrupted by another interruption management, but not by the management of the same interruption. The behavior is once again similar to those of signals.

We notice that we can register only one interruption catcher at a time, which may first appear as a restriction. In fact it is not, as it is shown in the following example.

The following program simulate a kind of *parallel run*. Indeed, on one hand, a small counter increments itself regularly on the display. On the other hand, a second counter is incremented at each electric impulse sent by the user on the B0 pin. Note that a short generic library of interruption management is defined.

#### OCaml source code

Place the following code in a `interrupt.ml` file:

```
1 open Pic;;
2
3 (* Speed-up of the PIC clock *)
4 set_bit IRCF1;; set_bit IRCF0;; set_bit PLEN;;
5
6 (* Configuration of the connection to the LCD display *)
7 module Disp = Lcd.Connect (
8   struct
9     let bus_size = Lcd.Four
10    let e = Pic.LATD0
11    let rs = Pic.LATD2
12    let rw = Pic.LATD1
13    let bus = Pic.PORTC
14  end
15 );;
16 open Disp;;
17
18 init ();;
19 config ();;
20
21 (* Short generic library for the management of interruptions *)
22
23 type interruption_behavior =
24   | Interruption_ignore
25   | Interruption_handle of (bit -> unit)
26 ;;
27
28 let handlers = ref [];;
29
30 let the_handler bit =
31   try (List.assq bit !handlers) bit
32   with Not_found -> ()
33 ;;
34
35 let interruption bit ib =
```

```

36 let old =
37   try
38     let old = Interruption_handle (List.assq bit !handlers) in
39     handlers := List.remove_assq bit !handlers;
40     old
41   with Not_found -> Interruption_ignore
42 in
43   match ib with
44   | Interruption_ignore -> old
45   | Interruption_handle f ->
46     set_interruption_handler the_handler;
47     handlers := (bit, f) :: !handlers ; old
48 ;;
49
50 let set_interruption bit ib = ignore (interruption bit ib);;
51
52 (* Using the short library *)
53
54 (* A specific handler *)
55 let my_handler =
56   let counter = ref 0 in
57   fun _ ->
58     let (li, co) = current_position () in
59     moveto 2 0;
60     print_string "_____";
61     moveto 2 0;
62     Printf.fprintf output "Interruption_%d" !counter;
63     incr counter;
64     moveto li co;
65     Sys.sleep 100;
66 ;;
67
68 (* Activation of the INT0 interruption for the PIC *)
69 set_bit INT0E;;
70 set_bit GIE;;
71
72 (* Registration of the handler *)
73 set_interruption INT0F (Interruption_handle my_handler);;
74
75 (* Regular increment of a counter *)
76 while true do
77   for i = 0 to max_int do
78     clear_bit GIE;          (* Beginning of a mutual exclusion area *)
79     moveto 1 0;            (* (to avoid display collision) *)
80     Printf.fprintf output "Loop_%d" i;
81     set_bit GIE;          (* End of the mutual exclusion area *)
82     Sys.sleep 100;
83   done
84 done

```

## Compilation

Compile interrupt.ml → interrupt, interrupt.asm :

```
ocapic 18f4620 config.asm interrupt.ml
```

Assemble interrupt.asm → interrupt.hex :

```
gpasm -y interrupt.asm
```

You should get the interrupt.hex file to transfer on the PIC. Connect an LCD display to the PIC as usual. While the program is running, you can send impulses to the B0 pin to increment the second counter, in parallel to the first one which increments regularly.

# Contents

<b>1</b>	<b>Introduction</b>	<b>2</b>
<b>2</b>	<b>Installation</b>	<b>2</b>
2.1	Prerequisites . . . . .	2
2.2	Downloading . . . . .	2
2.3	Installation . . . . .	2
<b>3</b>	<b>Use examples</b>	<b>3</b>
3.1	Blink a LED . . . . .	3
3.2	Hello world on an LCD display . . . . .	5
3.3	A small timer . . . . .	7
3.4	Communication with a computer . . . . .	10
<b>4</b>	<b>Advance use</b>	<b>13</b>
4.1	Virtual machine settings . . . . .	13
4.2	Binding with assembler code . . . . .	13
4.3	Interruption catching . . . . .	15